Computer Sciences Department
University of Wisconsin
1210 West Dayton Street
Madison, Wisconsin 53706

NUMERICAL STUDIES OF STEADY, VISCOUS, INCOMPRESSIBLE FLOW IN A CHANNEL WITH A STEP

by Donald Greenspan

APPENDIX
PROGRAMMING FLOWS IN A CHANNEL WITH A STEP

by M. McClellan

Computer Sciences Technical Report #17
March 1968

Rerun: July 1972

NUMERICAL STUDIES OF STEADY, VISCOUS, INCOMPRESSIBLE FLOW IN A CHANNEL WITH A STEP*

by

Donald Greenspan

1. INTRODUCTION.

In this paper we will apply a new digital computer technique to the study of two dimensional, steady, viscous, incompressible flow through a channel with a step. The method is vastly more economical and accurate than time dependent, step-ahead techniques. The power of the method is contained in the structure of the difference equations, which, for all Reynolds numbers $\Re$, yield diagonally dominant systems of linear algebraic equations [1]-[3].

2. THE GENERAL PROBLEM.

The initial problem to be considered will be formulated analytically as follows. Consider the channel with a step which is shown in Figure 2.1. Let S be the polygon ABCDEFGH and let R be the interior of S . On R the equations of motion to be satisfied are the two dimensional, Navier-Stokes equations, that is,

$$\Delta \psi = -\omega \tag{2.1}$$

$$\Delta \omega + \Re \left(\frac{\partial \psi}{\partial x} \frac{\partial \omega}{\partial y} - \frac{\partial \psi}{\partial y} \frac{\partial \omega}{\partial x}\right) = 0 , \tag{2.2}$$

*Funds for the computations described in this paper were made available by the Research Committee of the Graduate School of the University of Wisconsin.

where ψ is the stream function, ω is the vorticity, and $\Re$ is the Reynolds number. On S the boundary conditions to be satisfied are

$$\psi = 1 , \quad \frac{\partial \psi}{\partial y} = 0 , \quad \text{on HG} ; \tag{2.3}$$

$$\psi = 0 , \quad \frac{\partial \psi}{\partial y} = 0 , \quad \text{on AB, CD, EF} ; \tag{2.4}$$

$$\psi = 0 , \quad \frac{\partial \psi}{\partial x} = 0 , \quad \text{on BC, DE} \tag{2.5}$$

$$\psi = 3y^2 - 2y^3 , \; \omega = 12y - 6 , \quad \text{on AH} ; \tag{2.6}$$

and

$$\frac{\partial \psi}{\partial x} = 0, \quad \frac{\partial \omega}{\partial x} + \Re \frac{\partial \psi}{\partial y} (\omega + \frac{\partial^2 \psi}{\partial y^2}) = 0 , \quad \text{on FG} . \tag{2.7}$$

Conditions (2.6) are those of Poiseuille flow [4], while conditions (2.7), formulated by R. E. Meyer and communicated privately, make the flow horizontal and the pressure constant on FG .

3. THE NUMERICAL METHOD.

In this section we will describe in complete generality a numerical method for approximating solutions of boundary value problem (2.1) - (2.7). In the discussion, a reference appearing with a difference equation indicates where a derivation of the equation can be found. Particular examples and actual computations will be described in the next section.

For n a fixed positive integer, determine grid size h from $h = \frac{1}{n}$. Next, on and within polygon ABCDEFGH, construct and number in the usual way [1] the set of interior grid points R_h and the set of boundary grid points S_h. With regard to R_h and S_h, it will be assumed, with very little loss of generality, that h can be selected so that α_1, α_2, β, γ and δ are integral multiples of h.

We will aim at constructing on $R_h + S_h$ a pair of finite sequences of discrete functions

$$\psi^{(0)}, \psi^{(1)}, \psi^{(2)}, \ldots, \psi^{(k)}, \psi^{(k+1)} \tag{3.1}$$

$$\omega^{(0)}, \omega^{(1)}, \omega^{(2)}, \ldots, \omega^{(k)}, \omega^{(k+1)} \tag{3.2}$$

with the properties that, for some given tolerance ε,

$$|\psi^{(k)} - \psi^{(k+1)}| < \varepsilon \tag{3.3}$$

$$|\omega^{(k)} - \omega^{(k+1)}| < \varepsilon \tag{3.4}$$

at each point of $R_h + S_h$. All the functions in sequences (3.1) and (3.2) will be called outer iterates and the particular functions $\psi^{(k)}$ and $\omega^{(k)}$ will be taken to be approximations on $R_h + S_h$ to $\psi(x, y)$ and $\omega(x, y)$, respectively.

For the above purpose, we begin by defining $\psi^{(0)}$ and $\omega^{(0)}$ as follows. At each grid point in HG, set $\psi^{(0)} = 1$; at each grid point in ABCDEF set $\psi^{(0)} = 0$; at each grid point in AH determine $\psi^{(0)}$ from (2.6); and on the remaining grid points of $R_h + S_h$ determine $\psi^{(0)}$ by linear interpolation along the vertical grid lines. At each grid point in AH determine $\omega^{(0)}$ from (2.6) and on the remaining grid points of $R_h + S_h$ set $\omega^{(0)} = 0$.

The second element of sequence (3.1) is now determined as follows. At each grid point in HG set $\psi^{(1)} = 1$; at each grid point in ABCDEF set $\psi^{(1)} = 0$; and at each grid point in AH , determine $\psi^{(1)}$ from (2.6). Next, at each grid point (x, y) in R_h, write down the difference analogue [1]

$$(3.5)\quad -4\psi^{(1)}(x, y) + \psi^{(1)}(x+h, y) + \psi^{(1)}(x, y+h) + \psi^{(1)}(x-h, y) + \psi^{(1)}(x, y-h) = -h^2\,\omega^{(0)}(x, y)$$

of differential equation (2.1), while at each point (x, y) of S_h which is interior to FG , write down the difference analogue

$$(3.6)\qquad \psi^{(1)}(x, y) = \psi^{(1)}(x-h, y)$$

of the first condition in (2.7). One then solves the linear algebraic system generated by (3.5) and (3.6) by the generalized Newton's method [1] with over-relaxation factor r_ψ and denotes the solution by $\overline{\psi}^{(1)}$, which is of course defined only on R_h and on those points of S_h which are interior to FG . The function $\psi^{(1)}$ is then defined on this point set by the weighted average

$$(3.7)\qquad \psi^{(1)} = \rho\psi^{(0)} + (1-\rho)\overline{\psi}^{(1)}, \quad 0 \le \rho \le 1 ,$$

thus completing the definition of $\psi^{(1)}$ on all of $R_h + S_h$.

The second element of sequence (3.2) is now determined as follows. At each grid point in AH , determine $\omega^{(1)}$ from (2.6). At each grid point

(x, y) which is interior to HG approximate $\omega^{(1)}$ by [3]

$$(3.8) \qquad \overline{\omega}^{(1)}(x, y) = \frac{2}{h^2} - \frac{2}{h^2}\,\psi^{(1)}(x, y-h)\ ;$$

at each grid point (x, y) which is interior to AB, CD and EF approximate $\omega^{(1)}$ by

$$(3.9) \qquad \overline{\omega}^{(1)}(x, y) = -\frac{2}{h^2}\,\psi^{(1)}(x, y+h)\ ;$$

at each grid point (x, y) interior to BC approximate $\omega^{(1)}$ by

$$(3.10) \qquad \overline{\omega}^{(1)} = -\frac{2}{h^2}\,\psi^{(1)}(x-h, y)\ ;$$

at each grid point (x, y) interior to DE approximate $\omega^{(1)}$ by

$$(3.11) \qquad \overline{\omega}^{(1)}(x, y) = -\frac{2}{h^2}\,\psi^{(1)}(x+h, y)\ .$$

At the stagnation points B and E, merely set

$$(3.12) \qquad \omega^{(1)} = 0\ ,$$

while at F and G, which will never enter into the computations, we do not define $\omega^{(1)}$ at all.

At the points C and D, assume that (3.5) is valid with $\omega^{(0)}$ replaced by $\omega^{(1)}$, so that at C we approximate $\omega^{(1)}$ by

$$(3.13) \qquad \overline{\omega}^{(1)} = -\frac{1}{h^2}\,[\psi^{(1)}(x, y+h) + \psi^{(1)}(x-h, y)]$$

while at D we approximate $\omega^{(1)}$ by

(3.14) $$\overline{\omega}^{(1)} = -\frac{1}{h^2}\left[\psi^{(1)}(x+h, y) + \psi^{(1)}(x, y+h)\right]$$

Next, at each point (x, y) of R_h, proceed as follows [2]. Determine the values

(3.15) $$\mathcal{H} = \psi^{(1)}(x+h, y) - \psi^{(1)}(x-h, y)$$

(3.16) $$\mathcal{K} = \psi^{(1)}(x, y+h) - \psi^{(1)}(x, y-h)$$

and write down, as is appropriate, the following difference analogues of differential equation (2.2):

(3.17a) $$\left(-4 - \frac{\mathcal{H}\mathfrak{R}}{2} - \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y) + \omega^{(1)}(x+h, y) + \left(1 + \frac{\mathcal{H}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y+h)$$
$$+ \left(1 + \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x-h, y) + \omega^{(1)}(x, y-h) = 0, \quad (\mathcal{H} \geq 0,\ \mathcal{K} \geq 0),$$

(3.17b) $$\left(-4 - \frac{\mathcal{H}\mathfrak{R}}{2} + \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y) + \left(1 - \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x+h, y) + \left(1 + \frac{\mathcal{H}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y+h)$$
$$+ \omega^{(1)}(x-h, y) + \omega^{(1)}(x, y-h) = 0, \quad (\mathcal{H} \geq 0,\ \mathcal{K} < 0),$$

(3.17c) $$\left(-4 + \frac{\mathcal{H}\mathfrak{R}}{2} - \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y) + \omega^{(1)}(x+h, y) + \omega^{(1)}(x, y+h)$$
$$+ \left(1 + \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x-h, y) + \left(1 - \frac{\mathcal{H}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y-h) = 0, \quad (\mathcal{H} < 0,\ \mathcal{K} \geq 0),$$

(3.17d) $$\left(-4 + \frac{\mathcal{H}\mathfrak{R}}{2} + \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y) + \left(1 - \frac{\mathcal{K}\mathfrak{R}}{2}\right)\omega^{(1)}(x+h, y) + \omega^{(1)}(x, y+h)$$
$$+ \omega^{(1)}(x-h, y) + \left(1 - \frac{\mathcal{H}\mathfrak{R}}{2}\right)\omega^{(1)}(x, y-h) = 0, \quad (\mathcal{H} < 0,\ \mathcal{K} < 0).$$

In applying (3.17a)-(3.17d), the values of ω at boundary grid points not in GF are to be determined from (3.8)-(3.14). Finally, at each grid point (x, y) interior to FG , we write down the difference analogue

$$(3.18)\qquad \frac{\omega^{(1)}(x,y) - \omega^{(1)}(x-h,y)}{h} + \mathcal{R}\left[\frac{\psi^{(1)}(x,y+h) - \psi^{(1)}(x,y-h)}{2h}\right]\left[\omega^{(1)}(x,y) + \frac{\psi^{(1)}(x,y+h) - 2\psi^{(1)}(x,y) + \psi^{(1)}(x,y-h)}{h^2}\right] = 0$$

of the second condition in (2.7) .

One then solves the linear algebraic system generated by (3.17a)-(3.17d) and (3.18) by the generalized Newton's method with over-relaxation factor r_ω . This solution is denoted by $\overline{\omega}^{(1)}$.

To determine $\omega^{(1)}$ at those points of $R_h + S_h$ at which only $\overline{\omega}^{(1)}$ has been defined, we use the averaging formula

$$(3.19)\qquad \omega^{(1)} = \mu\,\omega^{(0)} + (1-\mu)\overline{\omega}^{(1)}, \quad 0 \le \mu \le 1,$$

thus completing the definition of $\omega^{(1)}$ on all of $R_h + S_h$.

The numerical method then proceeds by generating $\psi^{(2)}$ from $\omega^{(1)}$ just as $\psi^{(1)}$ was generated from $\omega^{(0)}$ and by generating $\omega^{(2)}$ from $\psi^{(2)}$ just as $\omega^{(1)}$ was generated from $\psi^{(1)}$. The indicated iteration is continued until, for some k , (3.3) and (3.4) are valid. Substitution of $\psi^{(k)}$ and $\omega^{(k)}$ into the difference approximations of (2.1) and (2.2) to assure that these are the desired solutions terminates the method.

4. EXAMPLES.

We shall now try to organize in a comprehensive way the large number of examples run on the CDC 3600 at the University of Wisconsin. Convergent results were obtained readily for $\varepsilon = 10^{-4}$, $h = \frac{1}{10}$, $\beta = 1$, $\delta = 1$, $\gamma = \frac{1}{2}$, $r_\psi = 1.8$, $r_\omega = 1.0$ as indicated in Table 4.1. All stream curves are plotted in Figures 4.1 - 4.6, while typical equivorticity curves are shown in Figures 4.7 - 4.9. A variety of checks were run to determine the validity of the results shown in Figures 4.1 - 4.9. For example, for $\Re = 100$ the case $\alpha_1 = 4$, $\alpha_2 = 20$, $\rho = 0.04$, $\mu = 0.7$ was run to verify that the channel length to the right had no effect on the size of the vortex; for $\Re = 200$ a more accurate solution was obtained with $\varepsilon = 10^{-6}$ to verify the existence of the vortex on the left; and for $\Re = 200$ Poiseuille conditions were assumed on FG in place of (2.7), the result being essentially the same as that obtained with (2.7). Other selected results have been organized in Table 4.2.

From Table 4.1 it can be seen that as the right vortex was increasing in size, the amount of computing time required for $\Re > 1000$ would have been exorbitant. Thus, for economy purposes and in order to study better the growth of the left vortex, a modified problem was studied as follows.

5. FLOW UP A STEP.

The channel shown in Figure 2.1 was modified by eliminating the downstream step, as shown in Figure 5.0. The analytical problem formulated in Section 2 was modified by asking that (2.7) now be satisfied on ED in

Figure 5.0 . The numerical approach to this new problem was essentially the same as that described in Section 3 . Convergent results were obtained readily for $\varepsilon = 10^{-4}$, $\alpha = 5$, $r_{\psi} = 1.8$ and $r_{\omega} = 1.0$. In the case $\Re = 2000$ with $h = \frac{1}{10}$, convergence was achieved with $\rho = 0.03$, $\mu = 0.3$ in 100 outer iterations in only 2 minutes 31 seconds of running time. In the case $\Re = 5000$ with $h = \frac{1}{10}$, convergence was achieved with $\rho = 0.03$, $\mu = 0.3$ in 100 iterations in only 2 minutes 26 seconds of running time. In the case $\Re = 10000$ with $h = \frac{1}{10}$, convergence was achieved with $\rho = 0.05$, $\mu = 0.7$ in 230 iterations in 5 minutes of running time. The streamlines and equivorticity curves for these cases are shown in Figures 5.1 - 5.6 .

$\Re$	α_1	α_2	ρ	μ	Approximate number of outer iterates	Approximate running time
10	4	4	0.04	0.7	30	6 minutes
50	4	4	0.04	0.7	60	8 minutes
100	4	4	0.04	0.7	100	10 minutes
200	4	4	0.04	0.7	150	12 minutes
500	4	10	0.03	0.85	470	50 minutes
1000	4	10	0.03	0.9	650	91 minutes

Table 4.1

$\Re$	α_1	α_2	h	ρ	μ	r_ψ	τ_ω	Convergent or Divergent
10	4	4	$\frac{1}{4}$	0	0	1.8	1.0	Divergent
10	4	4	$\frac{1}{4}$	0.1	0.5	1.8	1.0	Convergent
10^5	4	4	$\frac{1}{4}$	0.1	0.5	1.8	1.0	Divergent
10	4	4	$\frac{1}{8}$	0.1	0.1	1.8	1.0	Divergent
10	4	4	$\frac{1}{8}$	0.7	0.1	1.8	1.0	Divergent
10	4	4	$\frac{1}{8}$	0.7	0.7	1.8	1.0	Divergent
10^5	4	4	$\frac{1}{8}$	0.1	0.7	1.8	1.0	Divergent
10^5	4	4	$\frac{1}{8}$	0.1	0.3	1.8	1.0	Divergent
10^5	4	4	$\frac{1}{8}$	0.1	0.3	1.8	0.7	Divergent
10^5	4	4	$\frac{1}{8}$	0.1	0.7	1.8	0.7	Divergent
10	4	4	$\frac{1}{16}$	0.1	0.7	1.8	1.6	Divergent
10	4	4	$\frac{1}{10}$	0.1	0.4	1.9	1.0	Divergent
10	4	4	$\frac{1}{10}$	0.2	0.7	1.8	1.0	Divergent
100	4	4	$\frac{1}{8}$	0.1	0.1	1.8	1.3	Divergent
100	4	4	$\frac{1}{10}$	0.1	0.7	1.8	1.0	Convergent
500	4	4	$\frac{1}{10}$	0.07	0.7	1.8	1.0	Divergent
500	4	4	$\frac{1}{10}$	0.03	0.45	1.8	1.0	Divergent
500	4	4	$\frac{1}{10}$	0.04	0.6	1.8	1.0	Divergent
10^3	4	4	$\frac{1}{10}$	0.02	0.5	1.8	1.0	Divergent
10^3	4	4	$\frac{1}{10}$	0.1	0.5	1.8	1.0	Divergent
10^3	4	4	$\frac{1}{10}$	0.5	0.5	1.8	1.0	Divergent
10^3	4	4	$\frac{1}{10}$	0.5	0.9	1.8	1.0	Divergent
10^3	4	4	$\frac{1}{10}$	0.1	0.9	1.8	1.0	Convergent
10^4	4	4	$\frac{1}{10}$	0.05	0.8	1.8	1.0	Divergent
10^4	4	8	$\frac{1}{10}$	0.05	0.6	1.8	1.0	Divergent

Table 4.2



Figure 2.1



FIGURE 4.1
STREAMLINES FOR R=10



FIGURE 4.2
STREAMLINES FOR R=50



FIGURE 4.3
STREAMLINES FOR R=100



FIGURE 4.4
STREAMLINES FOR R = 200



FIGURE 4.5
STREAMLINES FOR R = 500



FIGURE 4.6
STREAMLINES FOR R = 1000



FIGURE 4.7 EQUIVORTICITY CURVES FOR R=10



FIGURE 4.8 EQUIVORTICITY CURVES FOR R=200



FIGURE 4.9 EQUIVORTICITY CURVES FOR R=1000



Figure 5.0



FIGURE 5.1. STREAMLINES FOR R=2000



FIGURE 5.2. STREAMLINES FOR R=5000



FIGURE 5.3. STREAMLINES FOR R=10000



FIGURE 5.4. EQUIVORTICITY CURVES FOR R = 2000



FIGURE 5.5. EQUIVORTICITY CURVES FOR R = 5000



FIGURE 5.6. EQUIVORTICITY CURVES FOR R = 10000

REFERENCES

1. D. Greenspan, Introductory Numerical Analysis of Elliptic Boundary Value Problems, Harper and Row, New York, 1965.

2. D. Greenspan, "Numerical Studies of Two Dimensional Steady State Navier-Stokes Equations for Arbitrary Reynolds Number," Technical Report #9, Department of Computer Sciences, University of Wisconsin, 1967.

3. D. Greenspan, "Numerical Studies of Viscous, Incompressible Flow for Arbitrary Reynolds Number," Technical Report #11, Department of Computer Sciences, University of Wisconsin, 1968.

4. H. Schlichting, Boundary Layer Theory, McGraw-Hill, New York, 1960.

APPENDIX

PROGRAMMING FLOWS IN A CHANNEL WITH A STEP

M. McClellan

Definitions of Main Program Variables and Parameters

PSI, W = stream function and vorticity vectors, resp.

XL ≤ XSL ≤ XSR ≤ XR = left end of channel, left end of step, right end of step, right end of channel, resp.

D = width of channel at step (D < 1.0, max width).

H = grid size.

M, N = number of vertical and horizontal lines in grid, resp.

OMEGAP = relaxation factor for PSI inner-iterations.

OMEGAW = relaxation factor for W inner-iterations.

RHO, MU = weights for PSI and W, resp.

TOLP = tolerance for PSI inner- and outer-iterations.

TOLW = tolerance for W inner- and outer-iterations.

TOLTEST = number of outer-iterations between tests for problem convergence.

TOLTESTP = number of PSI inner-iterations between tests for convergence.

TOLTESTW = number of W inner-iterations between tests for convergence.

ITERMAX = maximum number of outer-iterations for both PSI and W .

ITERMAXP = maximum number of inner-iterations for PSI .

ITERMAXW = maximum number of inner-iterations for W .

Program switches

PCONV = 1 , if PSI outer-iterations have converged.

TAPEUSE = 1 , if PSI and W are to be saved on tape.

INPTAPE = 1 , if PSI and W are to be initialized from tape.

RET = return address from W relaxation-coefficients block.

```
      PROGRAM NS2AB
      DIMENSION PSI(241,11),W(241,11),PSISAV(241,11),WSAV(241,11)
      COMMON/PARAMS/ M,N,M1,MSL,MSR,MR,MSLP1,NB,NST,MP,NP,XSL,XSR,XL,
     1 H,MS
      TYPE REAL MU,MU1
      TYPE INTEGER TOLTEST,TOLTESTP,TOLTESTW,RET
      TYPE LOGICAL PCONV,TAPEUSE,INPTAPE
C
C   READ INPUT.
      READ 901, NPROBS
 901  FORMAT(I5)
      DO 70 IPROB=1,NPROBS
      READ 904, XL,XSL,XSR,XR,D,H,OMEGAP,OMEGAW,RHO,MU,TOLP,TOLW,R
 904  FORMAT(10F5,2E5,E10)
      READ 906, MP,NP,ITERMAX,TOLTEST,ITERMAXP,TOLTESTP,
     1 ITERMAXW,TOLTESTW,TAPEUSE,INPTAPE
 906  FORMAT(8I5,2L5)
C
C   COMPUTE INITIAL PARAMETERS.
      XCH=XR-XL
      XCHL=XSL-XL
      XCHR=XR-XSR
      XS=XSR-XSL
      M=XCH/H+1.5
      MSL=XCHL/H+1.5
      MSR=(XSR-XL)/H+1.5
      MS=XS/H+1.5
      MR=XCHR/H+1.5
      MSLP1=MSL+1
      MSL1=MSL-1
      MSRP1=MSR+1
      MSR1=MSR-1
      N=1.0/H+1.5
      NB=(1.0-D)/H+1.5
      NBP1=NB+1
      NB1=NB-1
      NST=D/H+1.5
      M1=M-1
      N1=N-1
      NPTS=M*N-(MS-2)*NB1
      H2=H*H
      CP1=1.0-OMEGAP
      CP2=0.25*OMEGAP
      CW00=1.0-OMEGAW
      CW5=2./H2
      CW6=1.0/H2
      CW7=H2*OMEGAW
      R2=0.5*R
      CW=1.0
      RHO1=1.-RHO
      MU1=1.-MU
      RMAX=RMAXPSV=1.E+5
      ITER=ITOL=0
C
C   PRINT INITIAL PARAMETERS.
      PRINT 909, XCH,XCHL,XCHR,XS,CW,D,H,N,M,NPTS,OMEGAP,OMEGAW,
```

```
     1 TOLP,TOLW,TOLTEST,ITERMAX
 909  FORMAT(1H1,10X,13HPROBLEM NO. 2,10X,82HNAVIER-STOKES EQUATIONS FOR
     1 FLOW IN A CHANNEL WITH STEP (LPSTREAM-DOWNSTREAM CASE) //
     2 20X,19HLENGTH OF CHANNEL =,F8.4,31H  WITH LENGTH TO LEFT OF STEP
     2=,F8.4 / 49X,29HAND LENGTH TC RIGHT OF STEP =,F8.4 /
     2 58X,20HAND LENGTH OF STEP =,F8.4 /
     3 20X,22HWIDE PART OF CHANNEL =,F8.4 /
     4 20X,24HNARROW PART OF CHANNEL =,F8.4 /
     4 20X,15HGRID SIZE (H) =,F10.6 /
     5 20X,25HNO. OF HOROZONTAL LINES =,I5,28H AND NO. OF VERTICAL LINES
     6 =,I5  / 20X,19HTOTAL GRID POINTS =,I7 /
     7 20X,39HRELAXATION FACTOR FOR STREAM FUNCTION =, F6.2 /
     7 20X,33HRELAXATION FACTOR FOR VORTICITY =,F6.2 /
     8 20X,19HTOLERANCE FOR PSI =,E10.1 / 20X,17HTOLERANCE FOR W =,E10.1
     8 / 20X,43HTOLERANCE TEST CYCLE FCR OUTER-ITERATIONS =I5 /
     9 20X,30HMAXIMUM NUMBER OF ITERATIONS =,I6 )
      PRINT 9091, ITERMAXP,TOLTESTP,ITERMAXW,TOLTESTW,R,RHO,MU
 9091 FORMAT(20X,43HMAXIMUM ITERATIONS FOR PSI-INNER-ITERATIONS,I6 /
     1 20X,47HTOLERANCE TEST CYCLE FCR PSI INNER-ITERATIONS =,I6 /
     2   20X,41HMAXIMUM ITERATIONS FCR W-INNER-ITERATIONS,I6 /
     3 20X,45HTOLERANCE TEST CYCLE FCR W INNER-ITERATIONS =,I6 /
     4 20X,17HREYNOLDS NUMBER =,F10.2 /
     5 20X,25HWEIGHTING (RHO) FOR PSI =F6.2 /
     6 20X,22HWEIGHTING (MU) FOR W =,F6.2 //)
C
C   INITIALIZE VECTORS W AND PSI.
      IF(INPTAPE) 5,8
C   INITIALIZE FROM INPUT TAPE.
 5    REWIND 5
      DO 6 J=1,N
 6    READ (5) (PSI(I,J),I=1,M)
      DO 7 J=1,N
 7    READ (5) (W(I,J),I=1,M)
      GO TO 9
C   INITIALIZE BY STANDARD PROCEDURE.
 8    CALL INIT2AB(W,PSI)
C   PRINT INITIAL VECTORS W AND PSI.
 9    PRINT 911, ITER,RMAX
  911 FORMAT(///10X,16HAT ITERATION NO.,I6,20H  MAXIMUM RESIDUAL =,E12.4
     1 /20X,15HSTREAM FUNCTION)
      CALL PRTMAT(PSI)
      PRINT 912
 912  FORMAT(//20X,9HVORTICITY )
      CALL PRTMAT(W)
C
C
C   BEGIN MAIN LOOP.
C
C   TEST IF VECTORS TO BE SAVED ON TAPE.
 10   IF(TAPEUSE) 1003,101
 1003 P1=ITER/100
      P2=ITER/100.
      IF(P1 .NE. P2) GO TO 101
      REWIND 5
      DO 1005 J=1,N
 1005 WRITE (5) (PSI(I,J),I=1,M)
```

```
      DO 1006 J=1,N
 1006 WRITE (5) (W(I,J),I=1,M)
C
 101  ITER=ITER+1
      ITOL=ITOL+1
      DO 102 J=2,N1
      DO 102 I=2,M
 102  PSISAV(I,J)=PSI(I,J)
 105  RMAXP=1.E94
      ITERP=0
 106  ITOLP=0
 12   ITERP=ITERP+1
      ITOLP=ITOLP+1
      IF(ITOLP .LT. TOLTESTP) 15,25
C
C   SWEEP STREAM FUNCTION IN REGIONS TO LEFT AND RIGHT BELOW STEP.
 15   DO 20 J=2,NB
      DO 17 I=2,MSL1
 17   PSI(I,J)=CP1*PSI(I,J) +CP2*(PSI(I+1,J)+PSI(I,J+1)+PSI(I-1,J)
     1 + PSI(I,J-1)+H2*W(I,J))
      DO 20 I=MSRP1,M1
 20   PSI(I,J)=CP1*PSI(I,J) +CP2*(PSI(I+1,J)+PSI(I,J+1)+PSI(I-1,J)
     1 + PSI(I,J-1)+H2*W(I,J))
C   SWEEP STREAM FUNCTION IN REGION ABOVE STEP.
      DO 22 J=NBP1,N1
      DO 22 I=2,M1
 22   PSI(I,J)=CP1*PSI(I,J) +CP2*(PSI(I+1,J)+PSI(I,J+1)+PSI(I-1,J)
     1 + PSI(I,J-1)+H2*W(I,J))
C   COMPUTE STREAM FUNCTION ON RIGHT BOUNDARY, FG.
      DO 23 J=2,N1
 23   PSI(M,J)=PSI(M1,J)
      GO TO 12
C
 25   RMAX1P=0.0
C   SWEEP STREAM FUNCTION IN REGIONS TO LEFT AND RIGHT BELOW STEP,
C   AND COMPUTE RESIDUALS.
      DO 30 J=2,NB
      DO 27 I=2,MSL1
      PSIOLD=PSI(I,J)
      PSINEW=CP1*PSI(I,J) +CP2*(PSI(I+1,J)+PSI(I,J+1)+PSI(I-1,J)
     1 + PSI(I,J-1)+H2*W(I,J))
      PSI(I,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 26,27
 26   RMAX1P=RES
 27   CONTINUE
      DO 30 I=MSRP1,M1
      PSIOLD=PSI(I,J)
      PSINEW=CP1*PSI(I,J) +CP2*(PSI(I+1,J)+PSI(I,J+1)+PSI(I-1,J)
     1 + PSI(I,J-1)+H2*W(I,J))
      PSI(I,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 29,30
 29   RMAX1P=RES
 30   CONTINUE
C   SWEEP STREAM FUNCTION IN REGION ABOVE STEP, AND COMPUTE RESIDUALS.
```

```
      DO 31 J=NBP1,N1
      DO 31 I=2,M1
      PSIOLD=PSI(I,J)
      PSINEW=CP1*PSI(I,J) +CP2*(PSI(I+1,J)+PSI(I,J+1)+PSI(I-1,J)
     1 + PSI(I,J-1)+H2*W(I,J))
      PSI(I,J)=PSINEW
      RES=ABSF(PSINEW-PSIOLD)
      IF(RES .GT. RMAX1P) 305,31
 305  RMAX1P=RES
 31   CONTINUE
      RMAXP=RMAX1P
C   COMPUTE STREAM FUNCTION ON RIGHT BOUNDARY WITHOUT RESIDUALS.
      DO 315 J=2,N1
 315  PSI(M,J)=PSI(M1,J)
C
C   TEST PSI INNER-ITERATIONS FOR DIVERGENCE.
      IF(RMAXP.GT. 1.E+5 ) 32,35
 32   PRINT 9017, RMAXP,ITERP
 9017 FORMAT(//77H *****  DIVERGENCE IN PSI-INNER-ITERATIONS. PROBLEM AB
     1ANDONED. MAX RESIDUAL =E15.6,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
 9009 FORMAT(/ 20X,20HSTREAM FUNCTION, PSI )
      CALL PRTMAT(PSI)
      PRINT 9050
 9050 FORMAT(/ 20X,12HVORTICITY, W )
      CALL PRTMAT(W)
      GO TO 70
C   TEST PSI INNER-ITERATIONS FOR CONVERGENCE.
 35   IF(RMAXP.LE. TOLP) 40,45
 40   PRINT 915, ITERP,TOLP,RMAXP,ITER
 915  FORMAT( 26H *****  AT INNER-ITERATION,I6,10H TOLERANCE,E10.1,
     1 34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6,10H  FOR PSI(,I5,1H))
C   WEIGHT STREAM FUNCTION IN INTERIOR AND ON RIGHT BOUNDARY.
      DO 402 J=2,NB
      DO 401 I=2,MSL1
 401  PSI(I,J)=RHO*PSISAV(I,J)+RHO1*PSI(I,J)
      DO 402 I=MSRP1,M
 402  PSI(I,J)=RHO*PSISAV(I,J)+RHO1*PSI(I,J)
      DO 403 J=NBP1,N1
      DO 403 I=2,M
 403  PSI(I,J)=RHO*PSISAV(I,J)+RHO1*PSI(I,J)
      IF(ITOL.LT. TOLTEST) 50,405
 405  ITOL=0
      IF(ITER .EQ.1) GO TO 50
      RMAX1=0.0
C
C   COMPUTE OUTER-ITERATION RESIDUALS FOR STREAM FUNCTION.
      DO 42J=2,NB
      DO 414 I=2,MSL1
      RES=ABSF(PSI(I,J)-PSISAV(I,J))
      IF(RES .GT. RMAX1) 413,414
 413  RMAX1=RES
 414  CONTINUE
      DO 42 I=MSRP1,M
      RES=ABSF(PSI(I,J)-PSISAV(I,J))
```

```
      IF(RES .GT. RMAX1) 416,42
 416  RMAX1=RES
 42   CONTINUE
      DO 423 J=NBP1,N1
      DO 423 I=2,M
      RES=ABSF(PSI(I,J)-PSISAV(I,J))
      IF(RES .GT. RMAX1) 422,423
 422  RMAX1=RES
 423  CONTINUE
      RMAX=RMAXPSV=RMAX1
C
C   TEST PSI OUTER-ITERATIONS FOR DIVERGENCE.
      IF(RMAX .GT.1.E+5 ) 432,435
 432  PRINT 9432, ITER,RMAX
 9432 FORMAT(//56H *****  DIVERGENCE IN STREAM FUNCTION AT OUTER-ITERATI
     1ON,I6,20H  MAXIMUM RESIDUAL =,E12.4 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C   TEST PSI OUTER-ITERATIONS FOR CONVERGENCE.
C   IF CONVERGENCE, GO TO COMPUTE AND TEST VORTICITY.
 435  PCONV=0
      IF(RMAX .LE. TOLP) 440,448
 440  PRINT 9440, ITER,TOLP,RMAX
 9440 FORMAT(   26H   ***  AT OUTER-ITERATION,I6,14H PSI-TOLERANCE,E10.1
     1 ,34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6 /)
      PCONV=1
      GO TO 50
C
C   FOR VORTICITY
C   COMPUTE OUTER-ITERATION RESIDUALS EVERYWHERE BUT LEFT BOUNDARY, AH.
 4402 RMAX1=0.0
      DO 442 J=1,NB
      DO 441 I=2,MSL
      RES=ABSF(W(I,J)-WSAV(I,J))
      IF(RES .GT. RMAX1) 4405,441
 4405 RMAX1=RES
 441  CONTINUE
      DO 442 I=MSR,M
      RES=ABSF(W(I,J)-WSAV(I,J))
      IF(RES .GT. RMAX1) 4415,442
 4415 RMAX1=RES
 442  CONTINUE
      DO 443 J=NBP1,N
      DO 443 I=2,M
      RES=ABSF(W(I,J)-WSAV(I,J))
      IF(RES .GT. RMAX1) 4425,443
 4425 RMAX1=RES
 443  CONTINUE
      RMAX=RMAX1
C
C   TEST VORTICITY OUTER-ITERATIONS FOR DIVERGENCE.
      IF(RMAX .GT.1.E+5) 4432,4435
```

```
 4432 PRINT 9532, ITER,RMAX
 9532 FORMAT(//50H *****  DIVERGENCE IN VORTICITY AT OUTER-ITERATION,I6,
     1 20H  MAXIMUM RESIDUAL =,E12.4 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C   TEST VORTICITY OUTER-ITERATIONS FOR CONVERGENCE.
C   IF CONVERGENCE, AND IF PSI HAS CONVERGED, SOLUTION OBTAINED.
 4435 IF(RMAX .LE. TOLW) 4439,445
 4439 PRINT 9443, ITER,TOLW,RMAX
 9443 FORMAT(   26H    ***  AT OUTER-ITERATION,I6,12H W-TOLERANCE,E10.1
     1 ,34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6 /)
      IF(PCONV) 444,48
 444  MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
C   SAVE SOLUTION ON TAPE, IF REQUIRED.
 4441 IF(TAPEUSE) 4442,70
 4442 REWIND 5
      DO 4444 J=1,N
 4444 WRITE (5) (PSI(I,J),I=1,M)
      DO 4445 J=1,N
 4445 WRITE (5) (W(I,J),I=1,M)
      GO TO 70
C   TEST IF MAXIMUM OUTER ITERATIONS EXCEEDED.
 445  PRINT 9445, ITER,RMAX
 9445 FORMAT(26H    ***  AT OUTER-ITERATION,I6,34H  MAXIMUM RESIDUAL FOR
     1VORTICITY =,E15.6 /)
      IF(ITER .GE. ITERMAX) 447,48
 447  PRINT 913
 913  FORMAT(//65H *****  MAXIMUM NUMBER OF OUTER-ITERATIONS USED. ABAND
     1ON PROBLEM. )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 4441
C
 448  PRINT 9448, ITER,RMAXPSV
 9448 FORMAT(26H    ***  AT OUTER-ITERATION,I6,40H  MAXIMUM RESIDUAL FOR
     1STREAM FUNCTION =,E15.6 /)
      GO TO 50
C   TEST IF MAXIMUM PSI INNER-ITERATIONS EXCEEDED.
 45   IF(ITERP.GE. ITERMAXP) 47,106
 47   PRINT 9013, RMAXP,ITERP
 9013 FORMAT(// 88H *****  MAXIMUM NUMBER OF INNER-ITERATIONS USED FOR S
     1TREAM FUNCTION.  MAXIMUM RESIDUAL =E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
```

```
      PRINT 9050
      CALL PRTMAT(W)
      GO TO 70
C
 48   PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
      PRINT 9480
 9480 FORMAT(///)
      GO TO 10
C
C    BLOCK TO COMPUTE CW-COEFFICIENTS FOR VORTICITY.
 4800 A=PSI(I+1,J)-PSI(I-1,J)
      B=PSI(I,J+1)-PSI(I,J-1)
      R2A=R2*A
      R2B=R2*B
      IF(R2A .GE. 0.) 4851,4855
 4851 IF(R2B .GE. 0.) 4852,4853
 4852 CW0=4.0+R2A+R2B
      CW1=CW4=1.0
      CW2=1.+R2A
      CW3=1.0+R2B
      GO TO 4860
 4853 CW0=4.0+R2A-R2B
      CW3=CW4=1.0
      CW2=1.0+R2A
      CW1=1.0-R2B
      GO TO 4860
 4855 IF(R2B .GE. 0.) 4856,4857
 4856 CW0=4.0-R2A+R2B
      CW1=CW2=1.0
      CW3=1.0+R2B
      CW4=1.0-R2A
      GO TO 4860
 4857 CW0=4.0-R2A-R2B
      CW2=CW3=1.0
      CW1=1.0-R2B
      CW4=1.0-R2A
C
 4860 CW0=OMEGAW/CW0
      GO TO (525,53,54,639,6403,648), RET
C
 50   RMAXW=1.E91
C    SAVE VORTICITY OF PREVIOUS OUTER-ITERATION.
      DO 502 J=1,N
      DO 502 I=1,M
 502  WSAV(I,J)=W(I,J)
C    COMPUTE VORTICITY ON TOP AND BOTTOM BOUNDARIES FOR THIS OUTER-ITER.
C    THE LINE AB.
      DO 5021 I=2,MSL1
 5021 W(I,1)=-CW5*PSI(I,2)
C    THE LINE CD.
      DO 5022 I=MSLP1,MSR1
 5022 W(I,NB)=-CW5*PSI(I,NBP1)
C    THE LINE EF.
```

```
      DO 5023 I=MSRP1,M1
 5023 W(I,1)=-CW5*PSI(I,2)
C     THE LINES CB AND ED.
      DO 5024 J=2,NB1
      W(MSR,J)=-CW5*PSI(MSRP1,J)
 5024 W(MSL,J)=-CW5*PSI(MSL1,J)
C     THE LINE HG.
      DO 5025 I=2,M1
 5025 W(I,N)=CW5*(1.0-PSI(I,N1))
C     THE CORNER POINTS C AND D.
      W(MSR,NB)=-CW6*(PSI(MSRP1,NB)+PSI(MSR,NBP1))
      W(MSL,NB)=-CW6*(PSI(MSL,NBP1)+PSI(MSR1,NB))
C
C     BEGIN VORTICITY INNER-ITERATIONS.
      ITERW=0
 505  ITOLW=0
 506  ITERW=ITERW+1
      ITOLW=ITOLW+1
      IF(ITOLW .LT. TOLTESTW) 507,63
C     THE RECTANGULAR REGION BELOW THE STEP ON THE LEFT.
 507  RET=1
      DO 525 J=2,NB
      DO 525 I=2,MSL1
      GO TO 4800
 525  W(I,J)=CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
C     THE RECTANGULAR REGION BELOW THE STEP ON THE RIGHT.
      RET=2
      DO 53 J=2,NB
      DO 53 I=MSRP1,M1
      GO TO 4800
 53   W(I,J)=CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
C     THE RECTANGULAR REGION ABOVE THE STEP.
      RET=3
      DO 54 J=NBP1,N1
      DO 54 I=2,M1
      GO TO 4800
 54   W(I,J)=CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
C     COMPUTE VORTICITY ON RIGHT BOUNDARY.
      DO 56 J=2,N1
      B=PSI(M,J+1)-PSI(M,J-1)
      CW8=R2*B
 56   W(M,J)=CW00*W(M,J)+(OMEGAW/(1.+CW8))*(W(M1,J)-(CW8/H2)*(PSI(M,J+1)
     1-2.*PSI(M,J)+PSI(M,J-1)))
      GO TO 506
C
 63   RMAX1W=0.0
C     COMPUTE ONE SWEEP OF VORTICITY IN INTERIOR, AND RESIDUALS.
C     THE RECTANGULAR REGION BELOW THE STEP ON THE LEFT.
      RET=4
      DO 640 J=2,NB
      DO 640 I=2,MSL1
      GO TO 4800
 639  WOLD=W(I,J)
```

```
      WNEW   =CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
      W(I,J)=WNEW
      RES=ABSF(WNEW-WOLD)
      IF(RES .GT.RMAX1W) 6395,640
 6395 RMAX1W=RES
 640  CONTINUE
C    THE RECTANGULAR REGION BELOW THE STEP ON THE RIGHT.
      RET=5
      DO 641 J=2,NB
      DO 641 I=MSRP1,M1
      GO TO 4800
 6403 WOLD=W(I,J)
      WNEW   =CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
      W(I,J)=WNEW
      RES=ABSF(WNEW-WOLD)
      IF(RES .GT.RMAX1W) 6405,641
 6405 RMAX1W=RES
 641  CONTINUE
C    THE RECTANGULAR REGION ABOVE THE STEP.
      RET=6
      DO 650 J=NBP1,N1
      DO 650 I=2,M1
      GO TO 4800
 648  WOLD=W(I,J)
      WNEW   =CW00*W(I,J)+CW0*(CW1*W(I+1,J)+CW2*W(I,J+1)+CW3*W(I-1,J)
     1 + CW4*W(I,J-1))
      W(I,J)=WNEW
      RES=ABSF(WNEW-WOLD)
      IF(RES .GT.RMAX1W) 649,650
 649  RMAX1W=RES
 650  CONTINUE
C    COMPUTE VORTICITY ON RIGHT BOUNDARY WITH RESIDUALS.
      DO 652 J=2,N1
      B=PSI(M,J+1)-PSI(M,J-1)
      CW8=R2*B
      WOLD=W(I,J)
      WNEW   =CW00*W(M,J)+(OMEGAW/(1.+CW8))*(W(M1,J)-(CW8/H2)*(PSI(M,J+1)
     1-2.*PSI(M,J)+PSI(M,J-1)))
      W(I,J)=WNEW
      RES=ABSF(WNEW-WOLD)
      IF(RES .GT.RMAX1W) 6515,652
 6515 RMAX1W=RES
 652  CONTINUE
      RMAXW=RMAX1W
C
C    TEST VORTICITY INNER-ITERATIONS FOR DIVERGENCE.
      IF(RMAXW.GT. 1.E+5 ) 665,666
 665  PRINT 9665, RMAXW,ITERW
 9665 FORMAT(// 55H *****  DIVERGENCE IN W-ONLY ITERATIONS, MAX RESIDUAL
     1 =,E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
```

```
      CALL PRTMAT(W)
      GO TO 70
C     TEST VORTICITY INNER-ITERATIONS FOR CONVERGENCE.
C     IF CONVERGENCE, THEN WEIGHT AND GO TO TEST OUTER-ITERATIONS.
  666 IF(RMAXW .LE. TOLW) 67,675
   67 PRINT 9067, ITERW,TOLW,RMAXW,ITER
 9067 FORMAT( 26H *****  AT INNER-ITERATION,I6,10H TOLERANCE,E10.1,
     1 34H SATISFIED WITH MAXIMUM RESIDUAL =,E15.6,8H  FOR W(,I5,1H) /)
C     WEIGHT VORTICITY EVERYWHERE EXCEPT LEFT BOUNDARY.
      DO 672 J=1,NB
      DO 6715 I=2,MSL
 6715 W(I,J)=MU*WSAV(I,J)+MU1*W(I,J)
      DO 672 I=MSR,M
  672 W(I,J)=MU*WSAV(I,J)+MU1*W(I,J)
      DO 6725 J=NBP1,N
      DO 6725 I=2,M
 6725 W(I,J)=MU*WSAV(I,J)+MU1*W(I,J)
      IF(ITOL .EQ. 0) 4402,10
C     TEST IF MAXIMUM VORTICITY INNER-ITERATIONS EXCEEDED.
  675 IF(ITERW .GE.ITERMAXW) 677,505
  677 PRINT 9677 , RMAXW,ITERW
 9677 FORMAT(// 80H *****  MAXIMUM NUMBER OF ITERATIONS USED FOR W-INNER
     1-ITERATIONS. MAX RESIDUAL =,E12.4,8H AT ITER,I6 )
      MP=NP=1
      PRINT 9009
      CALL PRTMAT(PSI)
      PRINT 9050
      CALL PRTMAT(W)
C
C     END OF MAIN LOOP
   70 CONTINUE
      STOP
      END
```